

C 64 extern — Der Weg nach draußen

In diesem Teil des Kurses werden wir uns mit dem Control-Port 1 beschäftigen. Er eignet sich nicht nur zum Anschluß eines Joysticks. Was man noch alles mit ihm machen kann, wollen wir am Beispiel eines Lichtgriffels im Selbstbau zeigen.

Am Control-Port 1 des C 64 gibt es nur noch einen Anschluß, dessen Bedeutung wir noch nicht genau kennen. Gemeint ist Pin 6, der Light Pen-Anschluß, von dem wir im letzten Teil schon etwas gehört haben. Er hat zwei verschiedene Funktionen. Zum einen wird hier die Information über den Zustand des Feuerknopfes eingespeist. Zur anderen interessanten Anwendung kommen wir jetzt.

Vielleicht haben Sie schon einmal von einer Computeranlage gehört (oder eine gesehen), bei der man bestimmte Funktionen mit einem speziellen Stift, der mit dem Computer verbunden ist, direkt am Bildschirm anwählen kann. Solch einen Stift nennt man Lichtgriffel. Ein Lichtgriffel ist sehr einfach an den C 64 oder den VC 20 anzuschließen. Man kann sich den »Lightpen« sogar selbst bauen.

Im Innern eines Lichtgriffels steckt als wichtigstes Element ein Fototransistor, eine Fotodiode oder ähnliches. Dieser Fototransistor reagiert wie unser LDR aus der letzten Folge auf Licht. Das Licht wird vom Elektronenstrahl der Bildröhre erzeugt, wenn dieser auf die Mattscheibe trifft. Der Fototransistor an den Bildschirm gehalten, registriert den Elektronenstrahl als kurzen Lichtimpuls. Das ist auch der Grund, weshalb der Lichtgriffel nur an relativ hellen Bildschirmstellen funktioniert, denn nur hier entsteht ein ausreichender Lichtimpuls. Wenn vom Lichtgriffel ein Impuls an den Computer gegeben wird, prüft dieser, für welchen Punkt des Schirms er gerade das Signal an den Bildschirm geschickt hat. Mit Hilfe dieser Information kann er dann die X- und Y-Koordinate der derzeitigen Position des Lichtgriffels auf dem Bildschirm zur Verfügung stellen.

Diese beiden Werte können wir abfragen und für unsere Programme verwenden.

Lichtgriffel selbstgebaut

Bevor wir wieder etwas programmieren, wollen wir uns einen Lichtgriffel beschaffen und an unseren Computer anschließen. Für den Lichtgriffel benötigen wir, wie oben schon gesagt, einen Fototransistor. Es wurde der Typ TIL 78 verwendet. Alle Bauteile, die für diesen Kursteil benötigt werden, sind im Elektronikfachhandel erhältlich. Der LDR aus der letzten Folge reicht für unseren Lichtgriffel nicht aus, weil er zu langsam auf die Lichtschwankungen reagiert und deshalb keine brauchbaren Impulse liefert.

Wenn Sie schon in der glücklichen Lage sind, einen echten Lichtgriffel zu besitzen, können Sie die Versuche selbstverständlich auch mit diesem ausführen. Im folgenden soll jedoch nur auf den Fototransistor eingegangen werden.

Der TIL 78 hat zwei Beine. Sie heißen »Kollektor« und »Emitter«. Nun stellt sich die Frage, welches Bein welche Funktion hat? Es gibt drei Möglichkeiten, dies herauszufinden. Die erste ist für den Elektronikneuling wohl die beste: Sie fragen in dem Fachgeschäft, in dem Sie den Fototransistor kaufen. Zu Hause haben Sie dann zwei weitere Möglichkeiten: Die Seite des Fototransistorgehäuses, auf der sich der Kollektor befindet, ist etwas abgeflacht. Außerdem hat der Kollektor das kürzere Anschlußbein. Die zweite Erkennungsmethode ist allerdings sicherer als die dritte, weil irgend jemand das längere Bein zum kürzeren gemacht haben könnte!

```

100 REM *****
110 REM *
120 REM *      DEMO      ZUM
130 REM *      -----
140 REM *
150 REM *      LICHTGRIFFEL
160 REM *      -----
170 REM *
180 REM * BY      TOBIAS      NICOL
190 REM *
200 REM *      NEUWIESENSTRASSE 20
210 REM *
220 REM *      60000 FRANKFURT 71
230 REM *
240 REM *****
250 :
260 REM ***** INFORMATION *****
270 PRINT "{CLR,4DOWN}SOLL DER PUNKT JEWEI
LS
280 PRINT "{DOWN}WIEDER GELOESCHT WERDEN?{
DOWN}
290 PRINT "BITTE 'J' ODER 'N' EINGEBEN!
300 GET A$: IF A$ = "" THEN 300
310 IF A$ = "N" THEN L = 1
320 :
330 REM ***** INITIALISIERUNG *****
340 POKE 53280,0 : POKE 53281,1
350 PRINT "{CLR}" : A = 1024 : B = 40
360 C = 24 : D = 53267
370 :
380 REM ***** FARBSPEICHER SETZEN *****
390 FOR I = 55296 TO 56295
400 POKE I,0 : NEXT I
410 :
420 REM ***** WERTE EINLESEN *****
430 REM ***** UND STABILISIEREN *****
440 FOR I = 1 TO 5
450 X = X + PEEK (D)
460 Y = Y + PEEK (D+1)
470 NEXT I
480 X = INT (X/5) : Y = INT (Y/5)
490 :
500 REM ** IN KOORDINATEN UMWANDELN **
510 KX = INT ((X-50)/180*(B-1))
520 KY = INT ((Y-45)/190*(C))
530 :
540 REM ** KOORDINATEN KONTROLLIEREN **
550 IF KX < 0 THEN KX = 0
560 IF KY < 0 THEN KY = 0
570 IF KX > B-1 THEN KX = B-1
580 IF KY > C THEN KY = C
590 :
600 REM *** PUNKT SETZEN/LOESCHEN ***
610 IF L=0 THEN POKE A+XA+YA*B,32
620 POKE A+KX+KY*B,160
630 :
640 REM ** NEUE WERTE & RUECKSPRUNG **
650 XA = KX : YA = KY : X = 0 : Y = 0
660 GOTO 420
670 :
680 :
690 :
700 REM *****
710 REM *** AENDERUNGEN FUER VC-20 ***
720 REM *****
730 REM
740 REM SETZEN SIE DIESE ZEILEN EIN:
750 REM
760 REM 340 POKE 36879,24
770 REM 350 PRINT "{CLR}" : A=7680 : B=22
780 REM 360 C=22 : D=36870
790 REM 390 FOR I = 38400 TO 38905
800 REM 510 KX = INT ((X-55)/95*(B-1))
810 REM 520 KY = INT ((Y-38)/89*(C))
820 REM
830 REM WENN SIE DAS PROGRAMM ABSPEI-
840 REM CHERN WOLLEN, SOLLTEN SIE NOCH
850 REM DIE ZEILEN 270-290 ANPASSEN.

```

© 64'er

**Listing 1: Demo zum Lichtgriffel.
Bitte mit dem Checksummer
eingeben.**

Schließen wir den »Lightpen« an den Computer an. Schalten Sie diesen dazu aus. Der Light-Pen-Anschluß ist der einzige an den Control-Port des C 64, der nur einmal, und zwar am Port 1, vorhanden ist. Verbinden Sie den Emitter mit GND und den Kollektor mit Light Pen des Port 1. Das war's schon!

Zur Demonstration habe ich das Programm »Demo zum Lichtgriffel« (Listing 1) geschrieben. Es ist den Joystick-Demos sehr ähnlich. Auch hier werden Punkte auf den Bildschirm gemalt. Nur diesmal erscheinen die Punkte dort, wo Sie den Lichtgriffel auf den Bildschirm setzen. Sie können also den Bildschirm als Eingabeeinheit verwenden und direkt auf ihm zeichnen.

Geben Sie das Programm ein und starten Sie es. Als erstes werden Sie gefragt, ob die erscheinenden Punkte jeweils wieder gelöscht werden oder gesetzt bleiben sollen. Antworten Sie mit einem der angegebenen Buchstaben. Nach wenigen Sekunden, in denen der Farbspeicher schwarz initialisiert wird, erscheint irgendwo ein kleines Rechteck auf dem Bildschirm. Nehmen Sie den Fototransistor zur Hand und führen Sie ihn über die Mattscheibe. Der Punkt folgt dem Lichtgriffel über den Bildschirm.

Auch Lichtgriffel und Tastatur stören einander. Wenn Sie unkontrolliert auf den Tasten herumtippen, sehen Sie, daß der Punkt auf dem Bildschirm hin- und herspringt. Ebenso erscheinen an der Cursorposition wahllos Zeichen, wenn sich der Computer im Direktmodus befindet und Sie den Lichtgriffel über den Bildschirm bewegen. Diese Störungen treten beim VC 20 nicht auf; hier lassen sich Tastatur und Lichtgriffel synchron benutzen.

Sollte sich der Punkt auf dem Bildschirm nicht vom Fleck bewegen, so kann das mehrere Gründe haben: Wenn Sie keinen Fehler beim Abschreiben des Listings gemacht haben (sich der VC 20 in der Grundversion befindet) und der Lichtgriffel korrekt angeschlossen ist, kann nur der Bildschirm zu dunkel eingestellt sein. Drehen Sie den Helligkeitsregler etwas weiter auf; das müßte den Lichtgriffel zum Ansprechen bringen.

Eine Verstärker-Schaltung

Funktioniert das Programm immer noch nicht, wird der Anschluß des Lichtgriffels etwas komplizierter. Sie benötigen eine Verstärkerschaltung, die die zu schwachen Signale vom Lichtgriffel für den Computer ver-




Bild 1. Die Schaltung des Lichtgriffelverstärkers

Halbleiter

1 NE 555 DIL
1 BC 238 C
1 TIL 78

Widerstände

1 kΩ
1 10 kΩ
1 Trimpoti 10 MΩ

Kondensatoren

1 nF

Sonstiges

etwas Kabel
Control-Port-Stecker

Stückliste 1. Diese Teile benötigen Sie für den Lichtgriffelverstärker

stärkt. Eine solche Schaltung gibt Bild 1 wieder. Sie wird durch die im Schaltbild angegebenen Anschlüsse mit dem Computer verbunden und kann sowohl an den C 64 als auch an den VC 20 angeschlossen werden.

Die Schaltung ist zum Glück recht einfach, sollte jedoch sorgfältig aufgebaut werden, um Kurzschlüsse zu vermeiden. Sie sollten sich deshalb vielleicht von einem Hobby-Elektroniker helfen lassen, wenn Sie sich noch nie mit Elektronik beschäftigt haben. Die Teile, die Sie für den Verstärker benötigen, entnehmen Sie bitte der Stückliste 1. Die folgende Funktionsbeschreibung der Schaltung ist für den fortgeschrittenen Hobby-Elektroniker gedacht und kann übersprungen werden.

Zur Schaltung: Sie besteht im wesentlichen aus einem IC (Integrated Circuit = Integrierter Schaltkreis) des Typs NE 555. Der NE 555 ist in unserer Schaltung als Monoflop (Monostabile Kippstufe) geschaltet. Wenn das Monoflop durch einen negativen Impuls an Pin 2 des ICs an-

gestoßen wird, verbleibt es eine kurze Zeit in einem Arbeitszustand und fällt dann in einen Grundzustand zurück. Die Triggerimpulse (Pin 2 heißt Trigger) kommen von unserem an den Bildschirm gehaltenen Lichtgriffel. Er sendet einen Impuls, wenn er vom Elektronenstrahl der Bildröhre überstrichen wird. An Pin 3 (das ist der Ausgang des ICs) entstehen so die Impulse, die vom Transistor verstärkt, dem Light-Pen-Anschluß des Computers zugeführt werden.

Mit Hilfe dieser Verstärkerschaltung funktioniert der Lichtgriffel auch bei einem wesentlich dunkleren Bildschirm. Der optimale Arbeitspunkt der Schaltung wird mit dem 1-Mega-Ohm-Potentiometer eingestellt. Das ist beim C 64 besonders sorgfältig durchzuführen, weil sich sonst wieder einmal Probleme mit den normalen Computerfunktionen ergeben. Drehen Sie einfach solange langsam am Potentiometer, bis der Lichtgriffel ordnungsgemäß funktioniert. Am VC 20 arbeitet die Schaltung problemlos, jedoch muß auch hier der Arbeitspunkt eingestellt werden. Der 1-Kilo-Ohm-Widerstand ist ein Vorwiderstand zum Transistor. Die restlichen Bauteile stellen die Zeit ein, während der das Monoflop im Arbeitszustand verweilt, bis es wieder in den Grundzustand zurückfällt.

Sehen wir uns jetzt das Programm einmal genauer an. In Zeile 390/400 wird der Farbspeicher initialisiert, da die Punkte auf dem Bildschirm schwarz erscheinen sollen. Die Lichtgriffelposition wird vom Computer in X- und Y-Richtung jeweils mit einer Genauigkeit von neun Bit ermittelt. Die acht höherwertigen der neun Bits des X-Wertes stehen in der Speicherstelle 53267 (VC 20: 36870), die des Y-Wertes wieder in der darauffolgenden, 53268 (VC 20: 36871). Das Bit 0 des X/Y-Wertes kann nicht abgefragt werden, wodurch eine Auflösung von nur zwei Bildpunkten

zustande kommt. Das ist jedoch für »niederauflösende« Grafik völlig ausreichend, geschweige denn für Steuerfunktionen. In den Zeilen 440 bis 470 werden diese beiden Speicherzellen je fünfmal ausgelesen. Die erhaltenen Werte werden addiert und anschließend durch 5 geteilt (Zeile 480). Diese Mittelwertbildung dient zur Stabilisierung der Werte, denn der X-Wert schwankt ziemlich stark, während die Y-Koordinate recht konstant ist. Das ist seltsamerweise bei beiden Computern so, scheint also systembedingt zu sein.

In den Zeilen 510/520 werden die ausgelesenen Werte in Bildschirmkoordinaten umgewandelt. Wenn man einen Lichtgriffel in eigenen Programmen einsetzen will, muß man hier etwas experimentieren. Jeder muß für jedes Programm neu herausfinden, welche Werte die Koordinaten annehmen sollen (können) und wie er die Werte für dieses Programm am besten aufbereitet. Ab Zeile 540 folgt dann die durch die Programme der letzten Folge schon fast zum Standard gewordene Wertekontrolle, die bei diesem Programm wegen des schwankenden X-Wertes besonders wichtig ist. Die verbleibenden Zeilen dürften hinreichend bekannt sein.

Listing 2 zeigt ein weiteres Demonstrationsprogramm zum Lichtgriffel. Hier wird der Lichtgriffel nicht zum Zeichnen, sondern zum Auswählen von Menüpunkten benutzt. Wegen der größeren Genauigkeit des Y-Wertes wird dazu nur dieser verwendet. Es wird also nur die vertikale Richtung des Lichtgriffels ausgewertet. Sie können mit dem Programm einen der drei Tongeneratoren des C 64/VC 20 bestimmen, auf dem dann ein Ton erklingt; die Tonhöhe hängt vom ausgewählten Tongenerator ab. Den gewünschten Tongenerator bestimmen Sie, indem Sie das entsprechende Feld auf dem Bildschirm mit dem Fototransistor berühren.

In Zeile 470 wird das Lichtgriffel-Y-Register ausgelesen. Aufgrund des erhaltenen Wertes werden dann ab Zeile 490 der angewählte Tongenerator und die Tonhöhe bestimmt. Im C 64-Programm wird dabei nur auf eine direkte Berührung des angegebenen Kästchens in vertikaler Richtung reagiert, während beim VC 20 aus Platzgründen in den Programmzeilen größere Bereiche akzeptiert werden. Wird kein neuer Tongenerator angewählt, ist also T2 (=neuer Tongenerator) gleich T1 (=alter Tongenerator), wird durch Zeile 550 verhindert, daß die Generatoren neu gesetzt werden, weil dadurch eine Tonstörung entsteht.

Wenn Sie den Lichtgriffel öfter in selbstgeschriebenen Programmen einsetzen wollen, sollten Sie sich mit dem Fototransistor einen echten Licht-»Griffel« basteln. Nehmen Sie dazu einen Kugel- oder Filzschreiber und entfernen Sie alle »Innereien«. Löten Sie an die beiden Anschlüsse des Fototransistors zwei ausreichend lange Kabel an und isolieren Sie sie sorgfältig gegeneinander. Ein Kurzschluß kann zwar keinen Schaden anrichten, doch setzt er den Fototransistor außer Betrieb. Stecken Sie den Fototransistor jetzt von hinten in den Stift und befestigen Sie ihn vor der Minenöffnung. Je kleiner diese ist,

desto konstanter werden Ihre Werte anschließend sein, weil ein kleinerer Bildschirmarschnitt auf den Fototransistor einwirkt. Sie müssen jedoch aufpassen, daß nicht zu wenig Licht auf den Fototransistor fällt, da er sonst nicht mehr anspricht. In diesem Fall bohren Sie die Minenöffnung einfach etwas auf. Sie können auch mit dem Helligkeitsregler am Fernseher oder Monitor etwas korrigieren. Nach dem Anschluß an den Computer besitzen Sie dann einen richtigen Lichtgriffel.

Damit ist die Besprechung der Control-Ports vorerst abgeschlossen.

(Tobias Nicol/ev)

```

100 REM ***** <238>
110 REM * <159>
120 REM * T O N G E B E R M I T * <002>
130 REM * ----- * <237>
140 REM * <189>
150 REM * L I C H T G R I F F E L * <072>
160 REM * ----- * <011>
170 REM * <219>
180 REM * BY TOBIAS NICOL * <229>
190 REM * <239>
200 REM * NEUWIESENSTRASSE 20 * <202>
210 REM * <003>
220 REM * 6000 FRANKFURT 71 * <042>
230 REM * <023>
240 REM ***** <122>
250 : <226>
260 REM ***** BILDSCHIRMAUFBAU ***** <126>
270 POKE 53281,1 : POKE 53280,3 <029>
280 PRINT " {CLR,DOWN,2SPACE}BITTE WAELHEN <087>
    SIE!"
290 PRINT " {2SPACE}----- {DOWN <095>
    }"
300 FOR I = 1 TO 3 <145>
310 PRINT " {2SPACE}↑*5" <142>
320 PRINT " {2SPACE}↓ B TONGENERATOR"; I <092>
330 PRINT " {2SPACE}↗X {2DOWN}" <001>

```

```

340 NEXT I <170>
350 : <072>
360 REM ***** SID-CHIP EINSTELLEN ***** <176>
370 T1 = 4 : S1 = 54272 <238>
380 FOR I = 1 TO 10 <016>
390 READ A,B <186>
400 POKE SI+A,B <027>
410 NEXT I <240>
420 DATA 24,15, 0,207, 1,50, 7,207 <214>
430 DATA 8,100, 14,207, 15,150, 6,240 <138>
440 DATA 13,240, 20,240 <173>
450 : <172>
460 REM ** LIGHTPENPOSITION ABFRAGEN ** <095>
470 Y = PEEK (53268) <176>
480 : <202>
490 REM ***** POSITION AUSWERTEN ***** <183>
500 IF Y > 79 AND Y < 91 THEN T2 = 4 <094>
510 IF Y > 119 AND Y < 131 THEN T2 = 11 <041>
520 IF Y > 159 AND Y < 171 THEN T2 = 18 <174>
530 : <254>
540 REM **** TONGENERATOR TAUSCHEN **** <025>
550 IF T2 = T1 THEN 460 <136>
560 POKE SI+T1,0 <078>
570 POKE SI+T2,17 <183>
580 : <048>
590 REM ** NEUE WERTE & RUECKSPRUNG ** <024>
600 T1=T2 <119>
610 GOTO 460 <150>
620 : <088>
630 : <098>
640 : <108>
650 REM ***** <024>
660 REM *** AENDERUNGEN FUER VC-20 *** <233>
670 REM ***** <044>
680 REM <234>
690 REM LOESCHEN SIE 370-440! <209>
700 REM <254>
710 REM GEBEN SIE EIN: <101>
720 REM <018>
730 REM 270 POKE 36879,27 <160>
740 REM 370 T2 = 36874 : POKE 36878,15 <153>
750 REM 470 Y = PEEK (36871) <085>
760 REM 500 IFY< 63THENT2=36874:GOTO540 <101>
770 REM 510 IFY< 85THENT2=36875:GOTO540 <025>
780 REM 520 IFY<103THENT2=36876:GOTO540 <109>
790 REM 560 POKE T1,0 <158>
800 REM 570 POKE T2,(T2-36871)*50 <060>

```

© 64'er

Listing 2. Tongeber mit Lichtgriffel. Bitte mit dem Checksummer eingeben.

Sortieren mit dem Computer (5)

Ein ausführlicher Geschwindigkeitsvergleich aller bisher erarbeiteten Sortieralgorithmen soll die Spreu vom Weizen trennen. Und ein weiteres Sortieralgorithmus wird vorgestellt.

Entgegen der anderslautenden Vorhersage im letzten Teil unseres Kurses, soll heute noch einmal auf Sortieralgorithmen in Basic eingegangen werden. Dieser Entschluß wurde aus mehreren Gründen gefaßt.

Erstens trafen nach Erscheinen der letzten Ausgabe mehrere interessante Beiträge zum Thema Sortieren in der Redaktion ein. Zweitens soll noch ein-

mal ein grafischer Vergleich sämtlicher Routinen stattfinden, und drittens werden wir uns heute noch ein wenig mit den Vor- und Nachteilen der einzelnen Sortiermethoden beschäftigen.

Als erstes soll eine Verbesserung an den Mann gebracht werden. Heinrich Studer schickte der Redaktion einen Sortieralgorithmus, der dem schon bekannten Straight Selection stark

ähnelt, wegen seines einfacheren Aufbaus jedoch weniger Speicherplatz benötigt. Der Einfachheit halber soll dieser Sortieralgorithmus den Namen Minisort bekommen.

Wie Sie aus dem Flußdiagramm in Bild 1 sehen können, besteht Minisort aus zwei geschachtelten FOR-NEXT-Schleifen. In der äußeren Schleife wird jeweils der Reihe nach ein Element genommen und mit sämtlichen Elementen der inneren Schleife verglichen. Wird während eines Vergleichs festgestellt, daß das andere Element aus dem Restfeld kleiner als unser Vergleichswert ist, so werden beide Variablen vertauscht.

Ist die innere Schleife also einmal durchlaufen worden, so befindet sich das kleinste Element des Gesamtfeldes nun an erster Stelle.

Jetzt wird in der äußeren Schleife zum nächsten Element übergegangen, wobei dieses wiederum mit dem Restfeld verglichen wird... und so weiter, bis

man beim letzten Element angekommen ist. Es muß sich hierbei dann zwangsläufig um den größten Wert handeln.

Für Minisort (Listing 1) wie auch für unsere übrigen Sortieralgorithmen gilt, daß sie das Hauptprogramm für ihren Betrieb benötigen. Das Hauptprogramm dient der Erstellung der verschiedenen Arrays und der Anzeige der erstellten und sortierten Felder. In Listing 2 wurde auch diesmal wieder das Hauptprogramm abgedruckt, da es um ein paar Sachen erweitert wurde. Es ist jetzt mit der Option »von Hand erstellen« zusätzlich möglich, jeweils vorwärts und rückwärts sortierte Felder zu erstellen, um die »Testkandidaten« (sprich: Sortierprogramme) auf Herz und Nieren zu testen.

Um Ihnen jedoch einen großen Teil der Arbeit abzunehmen, habe ich für diese Folge unseres Kurses noch einmal sämtliche Sortierprogramme, die bisher besprochen wurden, in den drei Bereichen »Feld un-